
Kapitel 6

Logikprogrammierung mit
Constraints

Nachdem wir nun sowohl die “reine” Logikprogrammierung als auch ihre Implementierung
in der Sprache Prolog betrachtet haben, wollen wir uns zum Schluss mit einer wichtigen
Erweiterung der Logikprogrammierung, der sogenannten Logikprogrammierung mit Cons-
traints (“Constraint Logic Programming”, CLP) beschäftigen. In Abschnitt 6.1 führen wir
zunächst die Logikprogrammierung mit Constraints formal ein. Anschließend diskutieren
wir in Abschnitt 6.2, wie die Integration von Constraints in der Programmiersprache Prolog

gelöst wird. Weitere Literatur zu dem Thema ist beispielsweise [FA03, MS98].

6.1 Syntax und Semantik von Constraint-Logikpro-

grammen

Unter einem Constraint versteht man eine Einschränkung oder Bedingung. Solche Bedin-
gungen werden typischerweise als atomare Formeln formuliert. Die folgende Definition führt
diesen Begriff formal ein. Unser Ziel ist nachher, Logikprogramme über einer Signatur (Σ, ∆)
um Constraints über einer Teilsignatur (Σ′, ∆′) zu erweitern. Hierbei erweist es sich als sinn-
voll, das Gleichheits-Prädikatssymbol “=” sowie die speziellen Prädikatssymbole true und
fail gesondert zu behandeln.

Definition 6.1.1 (Constraint-Signatur und Constraints) Sei (Σ, ∆) eine Signatur
mit true, fail ∈ ∆0 und = ∈ ∆2. Seien Σ′ ⊆ Σ und ∆′ ⊆ ∆ Teilmengen der Signatur,
wobei ∆′ keines der Prädikatssymbole true, fail oder = enthält. Dann heißt (Σ, ∆, Σ′, ∆′)
Constraint-Signatur. Die atomaren Formeln aus At(Σ′, ∆′,V)∪At(Σ, {=},V)∪ {true, fail}
werden dann als Constraints über der Signatur (Σ, ∆, Σ′, ∆′) bezeichnet.

Bei einer Constraint-Signatur bestimmen wir also eine Teilmenge Σ′ der Funktions-
symbole und eine Teilmenge ∆′ der Prädikatssymbole, aus denen die Constraints gebildet
werden. Hierbei dürfen Prädikate aus ∆′ nur auf Terme angewendet werden, die keine
Funktionssymbole außer denjenigen aus Σ′ enthalten. Das Gleichheitszeichen = darf hin-
gegen auf beliebige Terme angewendet werden. Alle atomaren Formeln mit Prädikaten aus
∆′ ∪ {true, fail, =} heißen dann Constraints.

120 KAPITEL 6. LOGIKPROGRAMMIERUNG MIT CONSTRAINTS

Beispiel 6.1.2 Als Beispiel betrachten wir eine Contraint-Signatur (Σ, ∆, Σ′, ∆′), bei der
Σ′ und ∆′ Funktions- bzw. Prädikatssymbole zur Verarbeitung ganzer Zahlen enthalten:

Σ′
0 = ZZ

Σ′
1 = {−, abs}

Σ′
2 = {+,−, ∗, /, mod, min, max}

∆′
2 = {#>=, #=<, #=, #\=, #>, #<}

Man bezeichnet diese Mengen Σ′ und ∆′ auch als ΣFD und ∆FD. Hierbei steht “FD” für
Finite Domains. Falls f ∈ Σ1, so ergeben sich unter anderem folgende Constraints:

X + Y #> Z ∗ 3

max(X, Y ) #= X mod 2

f(X) + 2 = Y + Z

Um zu entscheiden, wann ein Constraint wahr ist, benötigt man darüber hinaus eine
Constraint-Theorie CT . Hierbei ist CT eine Menge von Formeln und ein Constraint ist
wahr, wenn es aus der Formelmenge CT folgt.

Definition 6.1.3 (Constraint-Theorie) Sei (Σ, ∆, Σ′, ∆′) eine Constraint-Signatur.
Falls CT ⊆ F(Σ′, ∆′,V) erfüllbar ist und nur geschlossene Formeln enthält, so bezeich-
nen wir CT als Constraint-Theorie.

Beispiel 6.1.4 Sei SFD = (ZZ, α) die Struktur mit den ganzen Zahlen als Träger und der
“intuitiven” Deutung der Funktions- und Prädikatssymbole. Es gilt also αn = n für alle
n ∈ ZZ. Weiterhin ist α+ die Additionsfunktion, etc. und α#> ist die Menge aller Zahlenpaare
(n, m) mit n > m, etc. Die naheliegende Constraint-Theorie zu der Signatur aus Bsp. 6.1.2
ist dann die Menge CTFD, so dass für alle geschlossenen Formeln ϕ ∈ F(ΣFD, ∆FD,V) gilt:

ϕ ∈ CTFD gdw. SFD |= ϕ

Man beachte, dass CTFD nicht nur unendlich, sondern nicht entscheidbar (und nicht
einmal semi-entscheidbar) ist. Dies liegt daran, dass wir auch Funktionen wie die Multi-
plikation in ΣFD aufgenommen haben. Würden wir uns nur auf die Addition beschränken
(man bezeichnet dies dann als Presburger Arithmetik), dann könnte man CTFD als endliche
Menge definieren. Mit anderen Worten, es existiert eine endliche Menge von Axiomen, aus
denen dann genau alle wahren Formeln über ganze Zahlen folgen.

Wir geben nun die Syntax und Semantik von Logikprogrammen mit Constraints an. Die
Syntax unterscheidet sich nicht von der Syntax normaler Logikprogramme. Wir gehen dabei
davon aus, dass Logikprogramme immer bereits die Klauseln zur Definition von true und =
enthalten und es keine Klausel zur Definition von fail gibt. Auf den linken Seiten anderer
Regeln dürfen die Prädikatssymbole der Constraints nicht auftreten. Außerdem dürfen die
Prädikatssymbole aus ∆′ nur auf Terme mit Funktionssymbolen aus Σ′ angewendet werden.

Definition 6.1.5 (Syntax von Logikprogrammen mit Constraints) Eine nicht–
leere endliche Menge P von definiten Hornklauseln über einer Constraint-Signatur (Σ, ∆,
Σ′, ∆′) heißt Logikprogramm mit Constraints über (Σ, ∆, Σ′, ∆′), falls {true} ∈ P und
{X = X} ∈ P und falls für alle anderen Klauseln {B,¬C1, . . . ,¬Cn} ∈ P gilt:

(a) Wenn B = p(t1, . . . , tm), dann gilt p /∈ ∆′ ∪ {true, fail, =}.

(b) Wenn Ci = p(t1, . . . , tm) und p ∈ ∆′, dann gilt tj ∈ T (Σ′,V) für alle 1 ≤ j ≤ m.

Ebenso muss die Bedingung (b) auch für alle Anfragen {¬C1, . . . ,¬Cn} gelten.

Beispiel 6.1.6 Wir betrachten wieder eine Constraint-Signatur (Σ, ∆, ΣFD, ∆FD). Ein
Beispiel für ein Logikprogramm mit Constraints ist dann die folgende Klauselmenge P.
Hierbei verwenden wir wieder die übliche Schreibweise für Logikprogramme (als Fakten
und Regeln anstelle von Klauseln).

fakt(0,1).

fakt(X,Y) :- X #> 0, X1 #= X-1, fakt(X1,Y1), Y #= X*Y1.

Man erkennt die Ähnlichkeit zu dem Programm zur Berechnung der Fakultät aus Ab-
schnitt 5.1:

fak(0,1).

fak(X,Y) :- X > 0, X1 is X-1, fak(X1,Y1), Y is X*Y1.

Nun definieren wir die Semantik der Logikprogrammierung mit Constraints. Analog zu
Abschnitt 4.1.1 und 4.1.2 werden wir dies sowohl auf deklarative als auch auf operatio-
nelle Weise tun. (Eine Fixpunkt-Semantik der Logikprogrammierung mit Constraints wäre
ebenfalls möglich.)

Die deklarative Semantik der Logikprogrammierung mit Constraints ist sehr nahelie-
gend. Bei der normalen Logikprogrammierung werden alle Instantiierungen einer Anfrage
als “wahr” betrachtet, die aus den Klauseln des Programms P folgen. Die Programmklau-
seln werden hier also als Axiome behandelt. Bei der Logikprogrammierung mit Constraints
werden die Axiome aus P einfach um die zusätzlichen Axiome CT erweitert.

Definition 6.1.7 (Deklarative Semantik eines Constraint-Logikprogramms) Sei
P ein Logikprogramm mit Constraints und CT eine dazugehörige Constraint-Theorie. Sei
G = {¬A1, . . . ,¬Ak} eine Anfrage. Dann ist die deklarative Semantik von P und CT
bezüglich G definiert als

D[[P , CT, G]] = {σ(A1 ∧ . . . ∧ Ak) | P ∪ CT |= σ(A1 ∧ . . . ∧ Ak), σ ist Grundsubstitution}.

Beispiel 6.1.8 Wir betrachten das Programm P aus Bsp. 6.1.6, die Constraint-Theorie
CTFD aus Bsp. 6.1.4 und die Anfrage G = {¬fakt(1, Z)}. Die einzige Grundsubstitution
mit P∪CTFD |= σ(fakt(1, Z)) ist σ(Z) = 1. Es gilt also D[[P , CT, G]] = {fakt(1, 1)}. Bei der
Anfrage G′ = {¬fakt(X, 1)} gilt P∪CTFD |= σ1(fakt(X, 1)) und P∪CTFD |= σ2(fakt(X, 1))
für die Substitutionen σ1(X) = 0 und σ2(X) = 1. Man erwartet daher bei der Anfrage

?- fakt(1,Z).

die Antwort Z = 1 und bei der Anfrage

?- fakt(X,1).

die Antworten X = 0 und X = 1.

Def. 6.1.7 zeigt deutlich, dass die normale Logikprogrammierung ein Spezialfall der
Logikprogrammierung mit Constraints ist. Offensichtlich entsprechen Logikprogramme mit
leerer Constraint-Theorie CT gerade den bislang betrachteten normalen Logikprogrammen.

Korollar 6.1.9 Sei P ein Logikprogramm mit Constraints über Σ′ = ∅ und ∆′ = ∅. Dann
gilt für alle Anfragen G: D[[P , ∅, G]] = D[[P , G]].

Um zu erklären, wie Logikprogramme mit Constraints ausgewertet werden, wollen wir
nun auch die prozedurale Semantik definieren. Dies ist jedoch nicht ganz so einfach wie
die deklarative Semantik. Das Problem ist, dass die Axiome in CT beliebige Formeln sein
können (und nicht unbedingt nur definite Hornklauseln). Ob eine Anfrage aus den Pro-
grammklauseln von P folgt, lässt sich mit binärer SLD-Resolution untersuchen. Dies ist bei
den Axiomen in CT so nicht möglich. Stattdessen gehen wir ja davon aus, dass wir eine Tech-
nik besitzen, die Folgerbarkeit aus CT automatisch überprüft. Das Problem besteht jetzt
darin, diese Technik (zur Überprüfung der Folgerbarkeit aus CT ) mit der SLD-Resolution
(zur Überprüfung der Folgerbarkeit aus P) zu verbinden. Die Idee hierzu besteht darin,
auch die SLD-Resolutionsschritte mit Hilfe von Constraints darzustellen. Auf diese Weise
erhält man eine einheitliche Darstellung beider Arten von Beweisschritten (den Schritten,
die SLD-Resolution mit Klauseln aus P durchführen und den Schritten, die Constraints
mit Hilfe der Constraint-Theorie CT lösen). Wir werden daher Unifikations-Informationen
explizit mit Hilfe von Gleichheiten als Constraints repräsentieren. Wie das folgende Beispiel
zeigt, könnte man diese Darstellung auch bei normalen Logikprogrammen ohne Constraints
verwenden.

Beispiel 6.1.10 Wir betrachten hierzu das folgende (reine) Logikprogramm aus Abschnitt
5.1.

add(X,0,X).

add(X,s(Y),s(Z)) :- add(X,Y,Z).

Bei der bisherigen Definition der prozeduralen Semantik bestand eine Konfiguration (G, σ)
aus der momentanen Anfrage G und der bereits berechneten Substitution. Die Berechnung

startete mit der leeren (oder “identischen”) Substitution ∅. Bei der Anfrage {¬add(s(0),
s(0), U)} ergab sich also

(¬add(s(0), s(0), U), ∅)
⊢P (¬add(s(0), 0, Z), {X/s(0), Y/0, U/s(Z)})
⊢P (2, {X ′/s(0), Z/s(0)} ◦ {X/s(0), Y/0, U/s(Z)}

︸ ︷︷ ︸

{X′/s(0), Z/s(0), X/s(0), Y/0, U/s(s(0))}

)

Aus der zum Schluss erhaltenen Substitution erhält man die Antwortsubstitution, indem
man sie auf die Variable U aus der ursprünglichen Anfrage einschränkt. Es ergibt sich also
die Antwortsubstitution {U/s(s(0))}.

Nun ist die Idee, die Unifikationen nicht durchzuführen, sondern anstelle der benötigten
Unifikatoren einfach nur die Unifikationsprobleme aufzusammeln. Hierbei steht “A = B” für
das Problem, die beiden atomaren Formeln A und B zu unifizieren. Die Konfigurationen
haben nun die Gestalt (G, CO), wobei CO eine Konjunktion von Unifikationsproblemen der
Form “A = B” ist. Man startet hierbei mit der leeren Konjunktion, d.h., mit der Formel
true, die stets wahr ist.

(¬add(s(0), s(0), U), true)

⊢P (¬add(X, Y, Z), add(s(0), s(0), U) = add(X, s(Y ), s(Z)))

⊢P (2, add(X, Y, Z) = add(X ′, 0, X ′) ∧ add(s(0), s(0), U) = add(X, s(Y ), s(Z)))

Die zum Schluss erhaltene Konjunktion von Unifikationsproblemen kann man natürlich
weiter vereinfachen. Sie ist im Endeffekt äquivalent zu der Bedingung

X ′ = s(0) ∧ Z = s(0) ∧ X = s(0) ∧ Y = 0 ∧ U = s(s(0)) (6.1)

Die entspricht also gerade der Substitution, die man bislang bei der prozeduralen Semantik
von Logikprogrammen erhalten hat.

Der Unterschied zur prozeduralen Semantik normaler Logikprogramme aus Def. 4.1.5 ist
also, dass wir nun nicht mehr Konfigurationen der Form (G, σ) betrachten, bei denen σ eine
Substitution ist. Stattdessen haben die Konfigurationen nun die Gestalt (G, CO). Hierbei
soll CO eine Konjunktion von Constraints sein. CO darf also Gleichungen zwischen Ter-
men enthalten. Das Gleichheits-Prädikatssymbol kann aber nicht auf zwei atomare Formeln
angewendet werden (d.h. man kann nicht A = B schreiben). Aus diesem Grund definieren
wir A = B als Abkürzung für eine entsprechende Konjunktion von Gleichheiten zwischen
Termen.

Definition 6.1.11 (Gleichheit von Atomen) Seien A und B Atome. Dann definieren
wir die Formel A = B wie folgt:

• A = B ist die Formel fail, falls A = p(. . .), B = q(. . .) mit p 6= q

• A = B ist die Formel true, falls A = B = p

• A = B ist die Formel s1 = t1 ∧ . . . ∧ sn = tn, falls A = p(s1, . . . , sn) und B =
p(t1, . . . , tn)

Beispiel 6.1.12 Mit der obigen Definition von A = B sehen die Rechenschritte von Bsp.
6.1.10 nun folgendermaßen aus:

(¬add(s(0), s(0), U), true)

⊢P (¬add(X, Y, Z), add(s(0), s(0), U) = add(X, s(Y ), s(Z))
︸ ︷︷ ︸

s(0)=X ∧ s(0)=s(Y )∧U=s(Z)

)

⊢P (2, add(X, Y, Z) = add(X ′, 0, X ′)
︸ ︷︷ ︸

X=X′ ∧Y =0∧Z=X′

∧ s(0) = X ∧ s(0) = s(Y ) ∧ U = s(Z))

Es ergibt sich zum Schluss also die folgende Konjunktion von Constraints:

X = X ′ ∧ Y = 0 ∧ Z = X ′ ∧ s(0) = X ∧ s(0) = s(Y ) ∧ U = s(Z) (6.2)

Um die hierdurch entstehenden Konjunktionen von Constraints CO zu vereinfachen,
kann man eine Funktion “simplify” verwenden, die Konjunktionen von Constraints in hierzu
äquivalente Konjunktionen von (einfacheren) Constraints überführt. Dabei ist zu überlegen,
was “Äquivalenz” hier bedeuten soll. Die bislang einzigen möglichen Prädikatssymbole in
CO sind true, fail und =. Man sollte bei der Vereinfachung von CO daher die Axiome über
true, fail und = beachten. Wir verlangen daher für die Funktion simplify, dass

{∀X X = X, true} |= ∀ (CO ↔ simplify(CO))

gilt. Zu jeder quantorfreien Formel ϕ mit den Variablen X1, . . . , Xn bezeichnet ∀ϕ den
Allabschluss von ϕ, d.h. die Formel ∀X1, . . . , Xn ϕ. Analog dazu ist ∃ϕ der Existenzabschluss
von ϕ, d.h. die Formel ∃X1, . . . , Xn ϕ.

In Bsp. 6.1.12 könnte dann simplify( (6.2) ) = (6.1) sein, da offensichtlich bereits

∀X X = X |= ∀X ′, X, Y, Z, U (6.2) ↔ (6.1)

gilt. Selbstverständlich kann man natürlich auch schon nach jedem Berechnungsschritt
“simplify” anwenden, um die zwischendurch erhaltenen Konjunktionen von Constraints
zu vereinfachen.

Man erkennt, dass anstelle der Unifikatoren nun also nur noch die Gleichheiten zwi-
schen den zu unifizierenden Termen aufgesammelt werden. Dass dies in der Tat zu einem
äquivalenten Ansatz führt, liegt daran, dass das Axiom “∀X X = X” sicher stellt, dass
zwei Terme nur dann als “gleich” betrachtet werden, wenn sie syntaktisch gleich sind. Die
Formel “∀X X = X” dient somit also zur “Axiomatisierung der Unifikation”. Dies wird
durch das folgende Lemma ausgedrückt, das wir benötigen werden, um die Äquivalenz der
deklarativen und der prozeduralen Semantik zu beweisen.

Lemma 6.1.13 (Gleichheit und Unifikation) Für alle Terme s, t, alle Atome A, B und
alle Substitutionen σ gilt:

(a) ∀X X = X |= σ(s = t) gdw. σ(s) und σ(t) syntaktisch gleich sind.

(b) {∀X X = X, true} |= σ(A = B) gdw. σ(A) und σ(B) syntaktisch gleich sind.

Beweis.

(a) Wir zeigen zunächst die Richtung von rechts nach links. Sei I eine beliebige Interpre-
tation, die Modell von ∀X X = X ist. Da σ(s) = σ(t) und somit I(σ(s)) = I(σ(t))
ist, folgt auch I |= σ(s) = σ(t).

Nun beweisen wir die Richtung von links nach rechts. Wir betrachten die Interpreta-
tion I = (T (Σ,V), α, β) mit αf = f , α= = {(r, r) | r ∈ T (Σ,V)} und β(X) = X. Hier
gilt also I(r) = r für alle Terme r und I |= r1 = r2 gilt gdw. r1 und r2 syntaktisch
gleich sind. Offensichtlich ist I Modell von ∀X X = X. Aus der Voraussetzung folgt
also, dass I |= σ(s = t) und somit σ(s) = σ(t).

(b) Falls A = p(. . .), B = q(. . .) mit p 6= q, so existiert kein σ, so dass σ(A) und σ(B)
syntaktisch gleich sind. Da A = B = fail ist, gilt daher auch {∀X X = X, true} 6|=
σ(A = B).

Falls A = B = p, so sind σ(A) und σ(B) für alle Substitutionen σ syntaktisch gleich.
Da in diesem Fall A = B = true ist, gilt hier auch {∀X X = X, true} |= σ(A = B).

Falls A = p(s1, . . . , sn) und B = p(t1, . . . , tn) ist, so gilt σ(A) = σ(B), falls σ(si) =
σ(ti) für alle 1 ≤ i ≤ n gilt. Nach (a) ist dies genau dann der Fall, wenn {∀X X =
X, true} |= σ(si = ti) für alle i gilt. Dies ist also gleichbedeutend mit {∀X X =
X, true} |= σ(A = B). 2

Da“{∀X X = X, true}” die Unifizierbarkeit axiomatisiert, kann man auf diese Weise
natürlich auch feststellen, wann zwei Terme nicht unifizierbar sind. Man sollte eine Kon-
figuration (G1, CO1) daher nur dann in eine neue Konfiguration (G2, CO2) überführen,
wenn die neue Bedingung CO2 unter diesen Axiomen erfüllbar ist, d.h. wenn {∀X X =
X, true} |= ∃ CO2. Auf diese Weise wird z.B. verhindert, dass in Bsp. 6.1.12 die erste Pro-
grammklausel direkt bei der Anfrage {¬add(s(0), s(0), U)} angewendet wird. Man würde
dann nämlich die Konjunktion s(0) = X ∧ s(0) = 0 ∧ U = X erhalten. Es gilt aber

{∀X X = X, true} 6|= ∃X, U s(0) = X ∧ s(0) = 0 ∧ U = X.

Um nun die prozedurale Semantik von Logikprogrammen mit Constraints zu definieren,
erweitern wir die in Bsp. 6.1.10 und 6.1.12 vorgestellte Idee wie folgt: Die zweite Komponen-
te CO einer Konfiguration kann nun nicht nur Constraints mit true, fail und = enthalten,
sondern es sind nun auch Constraints möglich, die mit Prädikatssymbolen aus ∆′ gebil-
det werden. Entsprechend muss man nun auch die Axiome CT zusätzlich zu den Axiomen
∀X X = X und true betrachten, wenn man die Erfüllbarkeit von CO untersucht und wenn
man CO mit “simplify” vereinfacht. Hierbei geht man davon aus, dass man ein Verfahren
zur Verfügung hat, um die Gültigkeit existenzquantifizierter Constraints zu entscheiden.
Mit anderen Worten, falls CO eine Konjunktion von Constraints ist, so geht man davon
aus, dass es entscheidbar ist, ob CT ∪ {∀X X = X, true} |= ∃ CO gilt. (In der Realität ist
dies natürlich bei Constraint-Theorien wie CTFD nicht der Fall. In Abschnitt 6.2 werden
wir daher diskutieren, wie man dieses Problem in der Praxis “löst”.)

Definition 6.1.14 (Prozedurale Semantik eines Constraint-Logikprogramms)
Sei P ein Logikprogramm mit Constraints und CT eine dazugehörige Constraint-Theorie.

• Eine Konfiguration ist ein Paar (G, CO), wobei G eine Anfrage oder die leere Klausel
2 ist und wobei CO eine Konjunktion von Constraints ist.

• Es gibt einen Rechenschritt (G1, CO1) ⊢P (G2, CO2) gdw. G1 = {¬A1, . . . ,¬Ak} mit
k ≥ 1 und eine der beiden folgenden Möglichkeiten (A) oder (B) zutrifft:

(A) Es gibt ein 1 ≤ i ≤ k, so dass Ai kein Constraint ist. Dann:

– existiert eine Programmklausel K ∈ P und eine Variablenumbenennung ν mit
ν(K) = {B,¬C1, . . . ,¬Cn} und n ≥ 0, so dass

∗ G1 und ν(K) keine gemeinsamen Variablen haben

∗ CT ∪ {∀X X = X, true} |= ∃ CO1 ∧ Ai = B

– G2 = {¬A1, . . . ,¬Ai−1,¬C1, . . . ,¬Cn,¬Ai+1, . . . ,¬Ak}

– CO2 = CO1 ∧ Ai = B

(B) Es gibt ein 1 ≤ i ≤ k, so dass Ai ein Constraint ist. Dann:

– CT ∪ {∀X X = X, true} |= ∃ CO1 ∧ Ai

– G2 = {¬A1, . . . ,¬Ai−1,¬Ai+1, . . . ,¬Ak}

– CO2 = CO1 ∧ Ai

• Eine Berechnung von P bei Eingabe von G = {¬A1, . . . ,¬Ak} ist eine (endliche oder
unendliche) Folge von Konfigurationen der Form

(G, true) ⊢P (G1, CO1) ⊢P (G2, CO2) ⊢P . . .

• Eine mit (2, CO) terminierende Berechnung, die mit (G, true) startet, heißt erfolg-
reich. Die berechneten Antwortconstraints sind simplify(CO), wobei CT ∪ {∀X X =
X, true} |= ∀ (CO ↔ simplify(CO)).

Damit ist die prozedurale Semantik von P bezüglich G definiert als

P [[P , CT, G]] = {σ(A1 ∧ . . .∧Ak) | (G, true) ⊢+
P (2, CO),

σ ist Grundsubstitution mit
CT ∪ {∀X X = X, true} |= σ(CO)}.

Es gibt also nun zwei Möglichkeiten für einen Berechnungsschritt, je nachdem ob das
ausgewählte Atom Ai aus der Anfrage ein Constraint ist oder nicht. Falls es kein Constraint
ist (Fall (A)), so findet wie bisher SLD-Resolution mit einer Programmklausel statt. Aller-
dings wird die benötigte Unifikation von Ai und dem Kopf B der Programmklausel nicht
durchgeführt, sondern man nimmt stattdessen die Constraints Ai = B mit auf. Hierbei muss
jeweils geprüft werden, ob die aufgesammelte Konjunktion von Constraints erfüllbar bleibt.
Falls Ai hingegen ein Constraint ist (Fall (B)), so wird Ai direkt in die aufgesammelte
Konjunktion von Constraints mit aufgenommen, sofern diese erfüllbar bleibt.

Beispiel 6.1.15 Wir betrachten wieder das Programm P aus Bsp. 6.1.6, die Constraint-
Theorie CTFD aus Bsp. 6.1.4 und die Anfrage G = {¬fakt(1, Z)}. Während Bsp. 6.1.8
zeigte, dass sich bei der deklarativen Semantik D[[P , CT, G]] = {fakt(1, 1)} ergibt, wollen
wir nun die prozedurale Semantik illustrieren. Hierbei wenden wir die Funktion “simplify”
nach jedem Rechenschritt an, um die erhaltene Konjunktion von Constraints zu vereinfa-
chen. Außerdem haben wir die Negationen vor den einzelnen Literalen weggelassen, um die
Lesbarkeit zu erhöhen. Man erhält

(fakt(1, Z), true)

⊢P (X#> 0, X1#= X − 1, fakt(X1, Y1), Y #= X ∗ Y1, true ∧ fakt(1, Z) = fakt(X, Y )
︸ ︷︷ ︸

X=1∧Z=Y

)

⊢P (X1#= X − 1, fakt(X1, Y1), Y #= X ∗ Y1, X#> 0 ∧ X = 1 ∧ Z = Y
︸ ︷︷ ︸

X=1∧Z=Y

)

⊢P (fakt(X1, Y1), Y #= X ∗ Y1, X1#= X − 1 ∧ X = 1 ∧ Z = Y
︸ ︷︷ ︸

X1=0∧X=1∧Z=Y

)

⊢P (Y #= X ∗ Y1, fakt(X1, Y1) = fakt(0, 1) ∧ X1 = 0 ∧ X = 1 ∧ Z = Y
︸ ︷︷ ︸

X1=0∧Y1=1∧X=1∧Z=Y

)

⊢P (2, Y #= X ∗ Y1 ∧ X1 = 0 ∧ Y1 = 1 ∧ X = 1 ∧ Z = Y
︸ ︷︷ ︸

Y =1∧X1=0∧Y1=1∧X=1∧Z=1

)

Das Ergebnis der abschließenden Simplifikation ist somit die folgende Formel CO:

Y = 1 ∧ X1 = 0 ∧ Y1 = 1 ∧ X = 1 ∧ Z = 1

Die einzige Grundsubstitution mit CT ∪ {∀X X = X, true} |= σ(CO) ist daher σ =
{Y/1, X1/0, Y1/1, X/1, Z/1}. Somit ergibt sich P [[P, CT, G]] = {fakt(1, 1)}.

Nun können wir die Äquivalenz der deklarativen und der prozeduralen Semantik für
Logikprogramme mit Constraints beweisen. Dies zeigt dann, dass eine Implementierung
der Logikprogrammierung mit Constraints wie in Def. 6.1.14 tatsächlich der gewünschten
(deklarativen) Semantik entspricht und dass so (aufgrund von Korollar 6.1.9) tatsächlich
auch die normale Logikprogrammierung korrekt implementiert wird, sofern die Contraint-
Theorie CT leer ist. Zum Beweis des folgenden Satzes gehen wir ähnlich wie im Beweis des
entsprechenden Satzes für normale Logikprogramme (Satz 4.1.8) vor.

Satz 6.1.16 (Äquivalenz der deklarativen und prozeduralen Semantik) Sei P ein
Logikprogramm mit Constraints und CT eine dazugehörige Constraint-Theorie. Weiter sei
G = {¬A1, . . . ,¬Ak} eine Anfrage. Dann gilt D[[P , CT, G]] = P [[P , CT, G]].

Beweis. Wir zeigen zuerst die Korrektheit der prozeduralen Semantik bezüglich der dekla-
rativen Semantik, d.h. P [[P, CT, G]] ⊆ D[[P , CT, G]]. Sei σ(A1 ∧ . . . ∧ Ak) ∈ P [[P, CT, G]].
Dann gibt es eine erfolgreiche Berechnung der Form

(G, true) ⊢P (G1, CO1) ⊢P (G2, CO2) . . . ⊢P (2, CO)

mit CT ∪ {∀X X = X, true} |= σ(CO).

Zu zeigen ist, dass dann auch P ∪ CT |= σ(A1 ∧ . . . ∧ Ak) gilt. Wir verwenden In-
duktion über die Länge l der Berechnung. Genauer ist l die Anzahl der ⊢P-Schritte in
der Berechnung. Es existiert also ein Atom Ai in der Anfrage G, das im ersten Schritt
(G, true) ⊢P (G1, CO1) in der Anfrage ersetzt bzw. weggelassen wird.

1. Fall: Ai ist kein Constraint

Dann existiert eine Programmklausel K ∈ P und eine Variablenumbenennung ν mit ν(K) =
{B,¬C1, . . . ,¬Cn} und n ≥ 0. Hierbei haben G und ν(K) keine gemeinsamen Variablen,
es gilt CT ∪ {∀X X = X, true} |= ∃ Ai = B und wir haben

G1 = {¬A1, . . . ,¬Ai−1,¬C1, . . . ,¬Cn,¬Ai+1, . . . ,¬Ak} und CO1 = Ai = B. (6.3)

Im Induktionsanfang ist l = 1. Dann ist G1 = 2 und somit i = k = 1, n = 0,
ν(K) = {B} (d.h., die Programmklausel K ist ein Faktum) und CO = CO1. Da CT ∪
{∀X X = X, true} |= σ(CO) gilt, folgt also CT ∪ {∀X X = X, true} |= σ(A1 = B) und
somit auch {∀X X = X, true} |= σ(A1 = B), da CT die Prädikatssymbole true, fail und =
nicht enthält. Nach Lemma 6.1.13 (b) folgt also, dass σ ein Unifikator von A1 und B ist,
d.h. σ(A1) = σ(B). Da außerdem B eine Klausel von P ist, gilt demnach P |= σ(B) bzw.
P ∪ CT |= σ(B) und somit auch P ∪ CT |= σ(A1).

Nun betrachten wir den Indukionsschritt l > 1. Für G1 wie in (6.3) ist dann offensichtlich
auch

(G1, true) ⊢P (G2, CO′
2) ⊢P . . . ⊢P (2, CO′)

eine Berechnung, wobei CO = Ai = B ∧ CO′ ist. Aus CT ∪ {∀X X = X, true} |= σ(CO)
folgt daher auch CT ∪ {∀X X = X, true} |= σ(CO′). Nach der Induktionshypothese ergibt
sich also

P ∪ CT |= σ(A1 ∧ . . . ∧ Ai−1 ∧ C1 ∧ . . . ∧ Cn ∧ Ai+1 ∧ . . . ∧ Ak).

Da P |= C1 ∧ . . . ∧ Cn → B gilt, folgt offensichtlich auch

P ∪ CT |= σ(A1 ∧ . . . ∧ Ai−1 ∧ B ∧ Ai+1 ∧ . . . ∧ Ak).

Da außerdem aus CT ∪ {∀X X = X, true} |= σ(CO) auch CT ∪ {∀X X = X, true} |=
σ(Ai = B) folgt (und somit {∀X X = X, true} |= σ(Ai = B)), erhält man aus Lemma
6.1.13 (b), dass σ(Ai) = σ(B) gilt. So ergibt sich

P ∪ CT |= σ(A1 ∧ . . . ∧ Ai−1 ∧ Ai ∧ Ai+1 ∧ . . . ∧ Ak).

2. Fall: Ai ist ein Constraint

Dann ist CT ∪ {∀X X = X, true} |= ∃ Ai und wir haben

G1 = {¬A1, . . . ,¬Ai−1,¬Ai+1, . . . ,¬Ak} und CO1 = Ai. (6.4)

Im Induktionsanfang ist l = 1. Dann gilt wieder G1 = 2 und somit i = k = 1 und CO =
CO1. Da CT∪{∀X X = X, true} |= σ(CO) gilt, folgt also CT∪{∀X X = X, true} |= σ(A1)
und somit auch P ∪ CT |= σ(A1), da P auch {∀X X = X, true} enthält.

Nun betrachten wir den Induktionsschritt l > 1. Für G1 wie in (6.4) ist dann offensicht-
lich auch

(G1, true) ⊢P (G2, CO′
2) ⊢P . . . ⊢P (2, CO′)

eine Berechnung, wobei CO = Ai ∧ CO′ ist. Aus CT ∪ {∀X X = X, true} |= σ(CO) folgt
daher auch CT ∪ {∀X X = X, true} |= σ(CO′). Nach der Induktionshypothese ergibt sich
also

P ∪ CT |= σ(A1 ∧ . . . ∧ Ai−1 ∧ Ai+1 ∧ . . . ∧ Ak).

Da aus CT ∪ {∀X X = X, true} |= σ(CO) auch CT ∪ {∀X X = X, true} |= σ(Ai) folgt,
ergibt sich

P ∪ CT |= σ(A1 ∧ . . . ∧ Ai−1 ∧ Ai ∧ Ai+1 ∧ . . . ∧ Ak),

da P auch {∀X X = X, true} enthält.
Nun zeigen wir die Vollständigkeit der prozeduralen Semantik bezüglich der deklarativen

Semantik, d.h. D[[P , CT, G]] ⊆ P [[P, CT, G]]. Sei σ(A1 ∧ . . . ∧ Ak) ∈ D[[P , CT, G]]. Dann
gilt P ∪ CT |= σ(A1 ∧ . . . ∧ Ak). Wir trennen nun die einzelnen Atome A1, . . . , Ak danach
auf, ob sie Constraints sind oder nicht. O.B.d.A. seien A1, . . . , Aj keine Constraints und
Aj+1, . . . , Ak seien Constraints (wobei 0 ≤ j ≤ k). Da P (außer den Klauseln {X = X}
und {true}) keine positiven Literale mit Prädikatssymbolen aus ∆′ ∪ {true, fail, =} enthält
und andererseits CT nur die Prädikatssymbole aus ∆′ enthält, ist P∪CT |= σ(A1∧. . .∧Ak)
äquivalent zu P |= σ(A1 ∧ . . .∧Aj) und CT ∪{∀X X = X, true} |= σ(Aj+1 ∧ . . .∧Ak). Wie
im Vollständigkeitsbeweis der prozeduralen Semantik für normale Logikprogramme (Satz
4.1.8) zeigt man, dass aus P |= σ(A1 ∧ . . . ∧ Aj) folgt, dass es die Berechnung

({¬A1, . . . ,¬Aj}, true) ⊢+
P (2, CO1)

gibt. Somit existiert also auch die Berechnung

({¬A1, . . . ,¬Ak}, true) ⊢+
P ({¬Aj+1, . . . ,¬Ak}, CO1).

Hierbei enthält CO1 die Gleichungen zwischen den zu unifizierenden Atomen. Da σ ein
Unifikator ist, folgt mit Lemma 6.1.13 (b), dass auch {∀X X = X, true} |= σ(CO1) gilt.
Weiter gilt auch

({¬Aj+1, . . . ,¬Ak}, CO1) ⊢
+
P (2, CO1 ∧ CO2),

wobei CO2 = Aj+1 ∧ . . .∧Ak. Wegen CT ∪ {∀X X = X, true} |= σ(Aj+1 ∧ . . .∧Ak) haben
wir nun also CT ∪ ∪{∀X X = X, true} |= σ(CO1 ∧ CO2). Damit folgt σ(A1 ∧ . . . ∧ Ak) ∈
P [[P , CT, G]]. 2

Um den Indeterminismus 2. Art (d.h. den Indeterminismus bei der Auswahl des Literals
aus der Anfrage) aufzulösen, geht man wieder genau wie bei der normalen Logikprogram-
mierung vor. Auch bei der Logikprogrammierung mit Constraints beschränkt man sich also
auf kanonische Berechnungen, d.h. auf Berechnungen, bei denen stets das linkeste Literal
der Anfrage gewählt wird. In Def. 6.1.14 wäre somit stets i = 1. Der Indeterminismus 1. Art
wird ebenfalls wieder dadurch aufgelöst, dass der SLD–Baum in Tiefensuche von links nach
rechts durchsucht wird. Dies bedeutet, dass Programmklauseln wieder in der Reihenfolge
von oben nach unten betrachtet werden.

Um die dadurch entstehenden SLD–Bäume zu illustrieren, betrachten wir noch einmal
das fak- und das fakt-Programm aus Bsp. 6.1.6.

Beispiel 6.1.17 Für das fak-Programm aus Bsp. 6.1.6 und die Anfrage “?- fak(X,1)”
ergibt sich der folgende SLD–Baum.

fak(X, 1)
{X/0}

u
u
u
u
u
u
u

XXXXXXXXXXXXXXXXXX

2 X > 0, X1 is X − 1, fak(X1, Y1), 1 is X ∗ Y1

Die erste Antwortsubstitution ist also X = 0. Wenn man anschließend “;” eingibt, so führt
dies jedoch zu einem Programmabbruch, da man das Beweisziel “X > 0” in Prolog nicht
stellen darf. Das Problem ist, dass das Prädikat > zwei voll instantiierte arithmetische
Ausdrücke als Argument erwartet. Das Programm fak ist also nicht bidirektional. Es eignet
sich zur Berechnung der Fakultät von vorgegebenen Zahlen (d.h. für Anfragen wie “?-
fak(1,Z)”), aber nicht dazu, zu einer vorgegebenen Zahl (wie 1) herauszufinden, ob es eine
Zahl X gibt, so dass die Fakultät von X gerade 1 ist.

Hier zeigt sich einer der Vorteile der Logikprogrammierung mit Constraints. Im Unter-
schied zu “>” ist das Prädikatssymbol “#>” tatsächlich bidirektional. Wir stellen daher nun
für das fakt-Programm aus Bsp. 6.1.6 die Anfrage “?- fakt(X,1)”. Bei den SLD-Bäumen
werden wir jetzt die Kanten jeweils mit der aufgesammelten Konjunktion CO von Cons-
traints beschriften, wobei wir diese (mit “simplify”) vereinfachen. Falls für die enstehende
Konjunktion CO nicht CT ∪ {∀X X = X, true} |= ∃ CO gilt, so wird der entsprechende
Kind–Knoten im SLD–Baum nicht gebildet. Es ergibt sich daher folgender SLD–Baum:

fakt(X, 1)
X=0

dddddddddddddddddddddddddddddddddd

Y =1

2 X#> 0, X1#= X − 1, fakt(X1, Y1), Y #= X ∗ Y1

Y =1∧X#>0

X1#= X − 1, fakt(X1, Y1), Y #= X ∗ Y1

Y =1∧X#>0∧X1#=X−1

fakt(X1, Y1), Y #= X ∗ Y1

Y =1∧X1=0∧Y1=1∧X=1

dddddddddddddddddddddddd

Y =1∧X#>0∧X1#=X−1

Y #= X ∗ Y1

Y =1∧X1=0∧Y1=1∧X=1

X1#> 0, X2#= X1 − 1, fakt(X2, Y2), Y1#= X ∗ Y2, Y #= X ∗ Y1

Y =1∧X#>0∧X1#=X−1∧X1#>0

2 X2#= X1 − 1, fakt(X2, Y2), Y1#= X ∗ Y2, Y #= X ∗ Y1

Y =1∧X#>0∧X1#=X−1∧X1#>0∧X2#=X1−1

fakt(X2, Y2), Y1#= X ∗ Y2, Y #= X ∗ Y1

Y =1∧X#>0∧X1#=X−1∧X1#>0∧X2#=X1−1
Y =1∧X2=0∧Y2=1∧X1=1∧X=2

eeeeeeeeeeeeeeeeeeeeeeeee

Y1#= X∗Y2,

Y #= X ∗ Y1

Y =1∧X2=0∧Y2=1∧X1=1∧ X=2∧Y1=2

.

.

.

Y #= X ∗ Y1

Der rechte Pfad in dem Baum ist unendlich. Das Problem ist hier, dass das Constraint
Y #= X∗Y1 aus der Anfrage nie in die Konjunktion CO von Constraints aufgenommen wird,
da immer das fakt(. . .)-Constraint links davon steht. Würde man stattdessen Y #= X ∗ Y1

mit in CO aufnehmen, so würde man den Pfad sofort erfolglos abschließen können, da aus

Y = 1 ∧ X#> 0 ∧ X1#= X − 1 ∧ X1#> 0

folgt, dass Y = 1 und X#> 1 sein müssen, was jedoch dem Constraint Y #= X ∗ Y1

widerspricht.

Aus einem ähnlichen Grund lässt sich der unterste Pfad mit dem Knoten “Y #= X ∗Y1”
nicht mehr fortsetzen, da “Y #= X ∗ Y1” den bisherigen Constraints

Y = 1 ∧ X2 = 0 ∧ Y2 = 1 ∧ X1 = 1 ∧ X = 2 ∧ Y1 = 2

widerspricht.

Stellt man dem Programm also die Anfrage “?- fakt(X,1)”, so erhält man zunächst
die Antworten X = 0 und X = 1, da die Antwortconstraints auf diejenigen eingeschränkt
werden, die die Variable X der Anfrage enthalten. Gibt der Benutzer danach noch einmal
“;” ein, so terminiert das Programm nicht mehr.

Um dies zu verbessern, empfiehlt es sich, die beiden letzten atomaren Formeln im Rumpf
der zweiten fakt-Regel zu vertauschen. Man erhält dann das Programm

fakt(0,1).

fakt(X,Y) :- X #> 0, X1 #= X-1, Y #= X*Y1, fakt(X1,Y1).

Die Anfrage “?- fakt(X,1)” führt nun zu folgendem SLD–Baum:

fakt(X, 1)
X=0

fffffffffffffffffffffff

Y =1

2 X#> 0, X1#= X − 1, Y #= X ∗ Y1, fakt(X1, Y1)

Y =1∧X#>0

X1#= X − 1, Y #= X ∗ Y1, fakt(X1, Y1)

Y =1∧X#>0∧X1#=X−1

Y #= X ∗ Y1, fakt(X1, Y1)

Y =1∧X#>0∧X1#=X−1∧ 1#=X∗Y1

fakt(X1, Y1)
Y =1∧X1=0∧Y1=1∧X=1

ffffffffffffffffffffff

Y =1∧X#>0∧X1#=X−1∧ 1#=X∗Y1

2 X1#> 0, X2#= X1 − 1, Y1#= X1 ∗ Y2, fakt(X2, Y2)

Der rechte Pfad lässt sich nun nicht weiter fortsetzen, da die dann entstehende Konjunktion
von Constraints

Y = 1 ∧ X#> 0 ∧ X1#= X − 1 ∧ 1#= X ∗ Y1 ∧ X1#> 0

widersprüchlich ist. Somit ist der SLD–Baum nun endlich und die Anfrage “?- fakt(X,1)”
führt zu den Antworten X = 0 und X = 1 und terminiert mit “No”, falls der Benutzer
anschließend noch einmal “;” eingibt.



132 KAPITEL 6. LOGIKPROGRAMMIERUNG MIT CONSTRAINTS

6.2 Logikprogrammierung mit Constraints in Prolog

Nun zeigen wir, wie die Logikprogrammierung mit Constraint in der Programmiersprache
Prolog realisiert ist. Zunächst erkennt man, dass die Prädikate “true” und “=” tatsächlich
mit den gleichen Klauseln wie in Def. 6.1.5 vordefiniert sind. In einem Prolog-Programm
muss aber natürlich zunächst einmal angegeben werden, welche Constraint-Theorie CT ver-
wendet werden soll. Hierdurch wird implizit auch festgelegt, welche Funktions- und Prädi-
katssymbole in den Mengen Σ′ und ∆′ enthalten sein sollen.

Wie andere Programmiersprachen auch bieten Prolog-Implementierungen meist ein Mo-
dul-System an. Einzeln Programm-Bibliotheken sind dann in unterschiedlichen Modulen
abgelegt. Das Prädikat use module dient dazu, die Prädikate eines Moduls zu importieren.
Um beispielsweise die Bibliothek mit den Prädikatssymbolen der Constraint-Theorie CTFD

aus Bsp. 6.1.2 zu importieren, muss das Prolog-Programm die folgende Direktive enthalten:

:- use_module(library(clpfd)).

Danach stehen die Symbole aus ΣFD und ∆FD zur Verfügung (vgl. Bsp. 6.1.2) und die
zugrundeliegende Constraint-Theorie ist CTFD. Man kann nun also in der Tat das fakt-
Programm aus Bsp. 6.1.6 schreiben und die Antworten auf die Anfragen “?- fakt(1,Z).”
und “?- fakt(X,1).” sind Z = 1 bzw. X = 0 und X = 1.

Ein Problem ist jedoch, dass die Folgerbarkeit aus den Axiomen der Constraint-Theorie
oftmals unentscheidbar ist. So ist es z.B. bei der Theorie CTFD nicht entscheidbar, ob für
eine Konjunktion von Constraints CO

CT ∪ {∀X X = X, true} |= ∃ CO (6.5)

gilt. Selbst bei anderen Constraint-Theorien, bei denen diese Frage entscheidbar ist, kann
das dazugehörige Entscheidungsverfahren so aufwendig sein, dass man es nicht sinnvoll in
der Implementierung der Logikprogrammierung mit Constraints einsetzen kann.

Aus diesem Grund werden in realen Implementierungen der Logikprogrammierung mit
Constraints meist Techniken verwendet, die die Frage (6.5) approximieren. Hierbei kann es
sein, dass die Approximation behauptet, dass (6.5) gilt, obwohl dies nicht der Fall ist. Im
Fall der Constraint-Theorie CTFD wird meist die sogenannte Pfadkonsistenz verwendet.

Definition 6.2.1 (Pfadkonsistenz) Sei CO = ϕ1 ∧ . . . ∧ ϕm eine Konjunktion von
Constraints mit ϕi ∈ At(ΣFD, ∆FD,V). Seien X1, . . . , Xn die Variablen in CO und sei-
en D1, . . . , Dn jeweils Teilmengen von ZZ. Wir sagen dass D1, . . . , Dn zulässige Domains
für die Variablen X1, . . . , Xn bzgl. CO sind, falls für alle Constraints ϕi mit 1 ≤ i ≤ m
und alle Variablen Xj mit 1 ≤ j ≤ n gilt: Für alle aj ∈ Dj existieren a1 ∈ D1, . . . , aj−1 ∈
Dj−1, aj+1 ∈ Dj+1, . . . , an ∈ Dn so dass CTFD |= CO[X1/a1, . . . , Xn/an]. Falls es zulässige
Domains D1, . . . , Dn gibt, die alle nicht-leer sind, so heißt CO pfadkonsistent.

Der wesentliche Unterschied zwischen der Pfadkonsistenz von CO und der “wirklichen
Konsistenz” (d.h. der Frage, ob CTFD |= ∃ CO gilt), ist, dass bei der Pfadkonsistenz die
Constraints separat voneinander betrachtet werden. In Prolog-Implementierungen geht man
dabei meist so vor, dass man zunächst alle Dj auf ZZ setzt. Anschließend durchläuft man
sukzessive alle Constraints und alle Variablen und reduziert die jeweiligen Domains. Dies

wird solange durchgeführt, bis sich die Domains Dj nicht mehr ändern. Als simplify(CO)
wird dann üblicherweise eine zu

X1 ∈ D1 ∧ . . . ∧ Xn ∈ Dn (6.6)

äquivalente Formel ausgegeben.

Beispiel 6.2.2 Betrachten wir die folgende Formel CO

X1#> 5 ∧ X1#< X2 ∧ X2#< 9

Am Anfang ist D1 = ZZ und D2 = ZZ. Wir betrachten zunächst den ersten Constraint
X1#> 5. Wir müssen nun alle Elemente aus D1 löschen, bei denen dieser Constraint nicht
erfüllbar ist. Dies führt zu D1 = {6, 7, . . .}.

Jetzt betrachten wir den zweiten Constraint. Wir beginnen mit der Variable X1 und
löschen alle Elemente a1 ∈ D1, für die es kein a2 ∈ D2 gibt, so dass a1 < a2 ist. Solche
Elemente gibt es allerdings nicht. Nun nehmen wir die Variable X2 und löschen alle Elemente
a2 ∈ D2, für die es kein a1 ∈ D1 gibt, so dass a1 < a2 gilt. Dies führt zu D2 = {7, 8, . . .}.

Jetzt untersuchen wir den dritten Constraint. Dieser führt zu D2 = {7, 8}. Man wieder-
holt nun die Untersuchungen der Constraints, bis sich nichts mehr ändert. Die Betrachtung
des zweiten Constraints für die Variable X1 ergibt noch D1 = {6, 7}. In der Tat erhält man
bei der Anfrage

?- X1 #> 5, X1 #< X2, X2 #< 9.

die Antwortconstraints

X1 in 6 .. 7, X2 in 7 .. 8

Diese Schreibweise mit dem vordefinierten Prädikat in ist gleichbedeutend zu

6 #=< X1, X1 #=< 7, 7 #=< X2, X2 #=< 8

Das Prädikat in kann auch vom Programmierer in Programmen und Anfragen verwendet
werden. Hierbei steht inf und sup für −∞ und ∞. Wir könnten unsere Anfrage also auch
wie folgt umformulieren:

?- X1 in 6 .. sup, X1 #< X2, X2 in inf .. 8.

Man erkennt, dass im Allgemeinen die Formel (6.6) nicht äquivalent zu CO ist, d.h.
simplify ist nun nicht mehr äquivalenzerhaltend. In der Tat würde in unserem Beispiel nun
auch die Lösung X1 = 7, X2 = 7 zugelassen werden, obwohl diese die ursprüngliche Anfra-
ge nicht erfüllt. Es gibt allerdings ein Prädikat label, mit dem man erzwingen kann, dass
Prolog die möglichen Lösungen nach und nach aufzählt. Hierbei muss label als Argument
die Liste der Variablen bekommen, für deren Werte man sich interessiert. Die Anfrage

?- X1 #> 5, X1 #< X2, X2 #< 9, label([X1,X2]).

liefert daher

X1 = 6, X2 = 7 ;

X1 = 6, X2 = 8 ;

X1 = 7, X2 = 8

Hierbei müssen aber die Domains für die Variablen im Argument von label tatsächlich
endlich sein. Ansonsten führt die Anfrage mit label zu einem Programmfehler. Dies würde
also z.B. bei der folgenden Anfrage auftreten:

?- X1 #> 5, X1 #< X2, label([X1,X2]).

Es existieren aber nicht nur Beispiele, bei denen die Simplifikation (6.6) “inkorrekt”
arbeitet. Es gibt sogar Beispiele, bei denen die Constraints pfadkonsistent sind, obwohl sie
widersprüchlich sind.

Beispiel 6.2.3 Das einfachste solche Beispiel ist

?- X1 #> X2, X1 #=< X2.

Die Überprüfung der Pfadkonsistenz zeigt, dass es für jeden Wert a1 ∈ D1 = ZZ einen Wert
a2 ∈ D2 = ZZ gibt, so dass der erste Constraint erfüllt ist und ebenso gibt es für jeden Wert
a2 ∈ D2 = ZZ einen Wert a1 ∈ D1 = ZZ gibt, so dass der erste Constraint erfüllt ist. Dasselbe
gilt für den zweiten Constraint. Somit ergibt sich als Antwort:

X1 in inf .. sup, X2 in inf .. sup

Zum Abschluss wollen wir zwei typische Beispiele für Programme betrachten, die sich
sehr gut als Logikprogramm mit Constraints formulieren lassen. Das erste Beispiel benutzt
wieder die Constraint-Theorie CTFD.

Beispiel 6.2.4 Wir programmieren das n-Damen-Problem (unser Program stammt im we-
sentlichen von [NM96]). Hierbei hat man ein Schachbrett der Größe n× n und das Ziel ist,
auf dem Schachbrett n Damen zu platzieren, die sich gegenseitig nicht schlagen können.
Dies bedeutet, dass es keine Zeile, keine Spalte und keine Diagonale geben darf, in der
mehr als eine Dame steht.

Wir repräsentieren die Positionen der Damen als Liste [x1, . . . , xn], wobei die Zahl xi die
Nummer der Zeile ist, in der die Dame der i-ten Spalte steht. Die Positionen der Damen sind
also (x1, 1), . . . , (xn, n). Die Liste [x1, . . . , xn] ist also eine Permutation der Zahlen [1, . . . , n].
Um herauszufinden, wie man n Damen auf einem n × n Brett platzieren kann, ruft man

?- queens(n,L).

auf. Die atomare Formel mit dem vordefinierten Prädikat length stellt sicher, dass die
Ergebnisliste L die Länge n hat. Das Prädikat ins aus der Bibliothek clpfd ist ähnlich
wie in, aber es stellt sicher, dass alle Elemente der Liste L aus dem vorgegebenen Bereich
stammen. Es gilt also “[x1, . . . , xn] ins 1 .. N” falls “xi in 1 .. N” für alle 1 ≤ i ≤
n gilt. Das vordefinierte Prädikat all different aus der Bibliothek clpfd stellt sicher,
dass alle Elemente von L paarweise verschieden sind. Hierbei stehen also die ins- und
all different-Formeln wieder für Abkürzungen für (naheliegende) Konjunktionen von
Constraints.

queens(N,L) :- length(L, N),

L ins 1 .. N,

all_different(L),

safe(L),

label(L).

safe([]).

safe([X|Xs]) :- safe_between(X, Xs, 1),

safe(Xs).

safe_between(X, [], M).

safe_between(X, [Y|Ys], M) :- no_attack(X, Y, M),

M1 #= M + 1,

safe_between(X, Ys, M1).

no_attack(X, Y, N) :- X+N #\= Y, X-N #\= Y.

Das Prädikat safe dient dazu, sicherzustellen, dass die Damen in der Liste L nicht auf
gleichen Diagonalen sind. Wir werden es gleich anschließend genauer erklären. Schließlich
dient label(L) dazu, die einzelnen Elemente von L auszugeben.

Die atomare Formel safe(L) stellt sicher, dass keine Dame in L eine Dame schlagen
kann, die rechts von ihr steht. Man benutzt hier das Hilfsprädikat safe between, wobei
safe between(X, L, M) wahr ist, falls die Dame in Zeile X keine Dame in den Spalten
aus L schlagen kann, sofern M der Abstand der Spalte von X zu der ersten Spalte aus L ist.
Daher trifft no attack(X, Y, N) zu, falls die Dame in Zeile X und die Dame in Zeile Y, die
N Spalten weiter rechts ist, nicht auf derselben Diagonale sind. Beispielsweise erhält man

?- queens(4,L).

L = [2, 4, 1, 3] ;

L = [3, 1, 4, 2]

Als nächstes Beispiel betrachten wir eine andere Constraint-Theorie für Constraints
über reellen Zahlen.

Beispiel 6.2.5 Wir behandeln nun eine Constraint-Signatur (Σ, ∆, Σ′, ∆′), bei der Σ′ und
∆′ Funktions- und Prädikatssymbole zur Verarbeitung reeller Zahlen enthalten. Um diese
Symbole von den (z.T. gleich lautenden) Symbolen aus Σ \ Σ′ und ∆ \ ∆′ zu unterschei-
den, nehmen wir diesmal nicht neue Namen wie “# >”, sondern wir vereinbaren, dass
alle Constraints mit Prädikaten aus ∆′ in geschweifte Klammern geschrieben werden. Die
Constraint-Theorie CTR enthält dann alle entsprechenden wahren Formeln über IR. Um die
entsprechende Bibliothek zu importieren, benötigt man die Direktive

:- use_module(library(clpr)).

Als Beispiel wollen wir nun ein Programm schreiben, das zur Zins- und Rückzahlungsbe-
rechnung bei Krediten verwendet werden kann [FA03]. Hierbei ist mortgage(D,T,I,R,S)

wahr, falls

• D der Betrag ist, den man als Kredit aufgenommen hat (Debt)

• T die Dauer (in Monaten) ist, seit der man den Kredit schon aufgenommen hat

• I der Zinssatz ist, den man pro Monat zurück zahlen muss (Interest)

• R die Rückzahlung ist, die man pro Monat leisten muss

• S der Betrag an Schulden ist, den man nach T Monaten noch hat

Das Programm lautet dann wie folgt:

mortgage(D, T, I, R, S) :- {T = 0, D = S}.

mortgage(D, T, I, R, S) :- {T > 0, T1 = T - 1, D1 = D + D * I - R},

mortgage(D1, T1, I, R, S).

Falls noch kein Monat vergangen ist, so sind die verbleibenden Schulden S gerade der Betrag
D, den man als Kredit aufgenommen hat. Ansonsten erhöhen sich aufgrund der Zinsen pro
Monat die Schulden um D * I und jeden Monat verringern sie sich um die Rückzahlung R.

Man kann also nun fragen, wie groß die Schulden nach 30 Jahren (d.h. 360 Monaten)
noch sind, wenn man ursprünglich 100000 Euro Kredit aufgenommen hatte, der Zinssatz
pro Monat 1 % beträgt und man jeden Monat 1025 Euro zurück zahlt.

?- mortgage(100000, 360, 0.01, 1025, S).

S = 12625.9

Man sieht, dass man inzwischen schon 360 * 1025 = 369000 Euro zurückgezahlt hat und
dennoch immer noch eine Restschuld von 12625.9 Euro hat. Dies demonstriert den Effekt
der Zinsen.

Aufgrund der Bidirektionalität der Logikprogrammierung (und der Tatsache, dass die
Prädikate in den Constraints (im Unterschied zu den eingebauten arithmetischen Prädika-
ten) tatsächlich bidirektional arbeiten), kann man in den Anfragen beliebige Argumente
vorgeben und andere berechnen lassen. So kann man z.B. fragen, wie hoch der Kredit denn
sein dürfte, damit man ihn bei diesen Konditionen nach 30 Jahren abbezahlt hat.

?- mortgage(D, 360, 0.01, 1025, 0).

D = 99648.8

Um zu fragen, wie lange wir bei unserem ursprünglichen Kredit von 100000 Euro noch
zurückzahlen müssen, könnte man folgende Anfrage stellen:

?- {S =< 0}, mortgage(100000, T, 0.01, 1025, S).

S = -807.964, T = 374.0 ;

S = -1841.04, T = 375.0 ;

S = -2884.45, T = 376.0 ;

S = -3938.3, T = 377.0 ;

...

Dies bedeutet, dass man also nach 374 Monaten eine “Schuld” von -807,964 Euro hätte.
Im letzten Monat muss man also keine volle Rate von 1025 Euro mehr zurückzahlen. (Die
weiteren Antworten sind zwar korrekt, aber nicht wirklich beabsichtigt.)

Wenn man wissen will, wie der Zusammenhang zwischen dem ursprünglichen Kredit D
und der monatlichen Rückzahlung R ist (falls man nach 30 Jahren seine Schuld abbezahlt
haben will), dann kann man folgende Anfrage stellen:

?- mortgage(D,360,0.01,R,0).

{R=0.0102861*D}

Diese Beispiele illustrieren, dass die Ergänzung der Logikprogrammierung um Cons-
traints tatsächlich eine sehr sinnvolle und nützliche Erweiterung der Logikprogrammierung
darstellt.


